
CHAPTER 5

Euler Products and Euler Sums

The preceding chapters dealt with problems of a purely analytic nature. We
now come to problems which are much more closely related to number theory. In
the present section, we study the numerical computation of sums and products over
the set P of prime numbers. Since this set is so complicated, this seems a priori
hopeless, until one recalls the fundamental property of the Riemann zeta function
for ℜ(s) > 1:

ζ(s) =
∑

n≥1

1

ns
=
∏

p∈P

(

1− 1

ps

)−1

.

Thus if we can express our computation in terms of values of ζ we may be in good
shape.

5.1. Euler Sums

First, let us show how to compute for ℜ(s) > 1

S(s) =
∑

p∈P

1

ps
.

The idea is the following: thanks to the above product formula for ζ(s), we have

log(ζ(s)) = −
∑

p∈P

log(1− 1/ps) =
∑

p∈P

∑

m≥1

1/(mpms)

=
∑

m≥1

∑

p∈P

1/pms =
∑

m≥1

S(ms)/m .

Now recall the second Möbius inversion formula: if f(n) =
∑

m≥1 g(mn), then

g(n) =
∑

m≥1 µ(m)f(mn). The proof of this is immediate:
∑

m≥1

µ(m)f(mn) =
∑

m≥1

µ(m)
∑

ℓ≥1

g(ℓmn) =
∑

N≥1

g(Nn)
∑

m|N

µ(m) = g(n)

by the basic property of the Möbius function.
By what we have seen above we have log(ζ(ns))/n =

∑

m≥1 S(mns)/(mn), so

applying Möbius inversion to f(n) = log(ζ(ns))/n and g(m) = S(ms)/m we deduce
that S(ns) = n

∑

m≥1 µ(m) log(ζ(mns))/(mn), so that

S(s) =
∑

m≥1

µ(m)

m
log(ζ(ms)) .

Since z(M) = 1 + O(2−M ) the convergence of this series is at least in O(2−m),
which is not bad. But we can do much better: it is clear that the proof of the

254 5. EULER PRODUCTS AND EULER SUMS

formula obtained above gives more generally

S>N (s) =
∑

m≥1

µ(m)

m
log(ζ>N (ms)) ,

where S>N means that we restrict the sum to primes p > N , and ζ>N that we
take the Euler product for zeta restricted to primes p > N (which is of course not

the same as restricting the sum defining ζ to n > N). We have thus proved the
following:

Proposition 5.1.1. Set S(s) =
∑

p∈P p−s. For any N > 0 and s such that

ℜ(s) > 1 we have

S(s) =
∑

p∈P, p≤N

1

ps
+
∑

m≥1

µ(m)

m
log(ζ>N (ms)) ,

where

ζ>N (s) = ζ(s)
∏

p∈P, p≤N

(1 − 1/ps) .

The whole point of introducing the parameter N is that log(ζ>N (ms)) =
O(N−ms), so the series will converge much faster. Note, however, that N must
not be chosen too large because in that case ζ>N would be extremely close to 1 and
there may be cancellation errors. In practice we advise choosing 10 < N < 100 for
instance.

Note in passing that the above proposition proves that S(s) can be analytically
continued to the half-plane ℜ(s) > 0 with the rational points 1/k for k ∈ Z≥1

removed, where S(s) has logarithmic singularities, and it is easy to show that it
has a natural boundary ℜ(s) = 0. To give a random example, we compute in the
same way that

S(1/
√
2) = 0.33542796189207260941184354829777539924+ iπ .

For the following, we recall that the valuation of a power series is its order at
x = 0:

Corollary 5.1.2. Let A(x) =
∑

m≥1 a(m)xm be a power series with no con-

stant term with nonzero radius of convergence r and valuation v, and set S(A; s) =
∑

p∈P A(1/ps). Define

c(n) =
∑

d|n

µ(d)

d
a
(n

d

)

.

Then for all N ≥ 1 and ℜ(s) > max(− log(r)/ log(N), 1/v) we have

S(A; s) =
∑

p∈P, p≤N

A(1/ps) +
∑

n≥1

c(n) log(ζ>N (ns)) .

In particular, if a(1) = 0 and r > 1/2 we have

S(A; 1) =
∑

p∈P, p≤N

A(1/p) +
∑

n≥2

c(n) log(ζ>N (n)) .

Proof. Simply write S>N (A; s) =
∑

m≥1 a(m)S>N (ms) and use the proposi-
tion. The details are left to the reader. �

Since it is not immediate to estimate the speed of convergence in the general
case, we restrict to the case of a rational function F . If we write F = N/D with N
and D coprime polynomials, the expansion at infinity of F is the expansion at 0 of
F (1/x) = N(1/x)/D(1/x), so the radius of convergence r is equal to the modulus
of the zero of D(1/x) closest to the origin, hence 1/r is equal to the modulus of
the zero of D(x) furthest from the origin, and this is approximately given by the
command polrootsbound which gives an upper bound for this largest modulus,
which we will use since it is considerably faster than computing the roots using
polroots.

In addition, note that when computing log(ζN (s)) by the evident formula given
above, there is considerable cancellation, so it is necessary to increase the working
accuracy. All this is done in the following program:

SumEulerrat.gp

1 LogzetaN(s, N) =

2 { my(B = getlocalbitprec());

3

4 B += ceil(abs(real(s) - 1) * exponent(2*N));

5 localbitprec(B);

6 if (bitprecision(s) < B, s = bitprecision(s, B));

7 log(zeta(s) * prodeuler(p = 2, N, 1 - p^(-s)));

8 }

9

10 Sdmob(ser, n) =

11 sumdiv(n, d, moebius(d) * polcoeff(ser, n/d) / d);

12

13 /* Compute sum_{p >= a, prime} F(p^s), F rational function. */

14 SumEulerrat(F, s = 1, a = 2) =

15 { my(B = getlocalbitprec(), vx = variable(F), rs, N, r, lim);

16 my(FI, vF, sal, S);

17

18 FI = subst(F, vx, 1/vx); vF = valuation(FI, vx);

19 rs = real(s);

20 if (rs <= 1 / vF, error("real(s) <= 1/v"));

21 localbitprec(32);

22 r = 1 / max(polrootsbound(denominator(F)), 1);

23 N = ceil(max(30, a+1));

24 if (rs <= -log(r)/log(N), error("real(s) too small"));

25 lim = ceil(B * log(2) / log(N^rs * r)) + 1;

26 localbitprec(B);

27 sal = bitprecision(1., B + 32)*FI + O(vx^(lim+1));

28 bernvec(floor((lim * s + 1) / 2)); /* cache the B_{2n} */

29 S = sum(n = vF, lim,

30 my(m = Sdmob(sal, n)); if (m, m * LogzetaN(n*s, N)));

31 forprime (p = a, N, S += subst(F, vx, p^s));

32 return (bitprecision(S, B));

33 }

Examples (all essentially instantaneous):

? SumEulerrat(1 / x^2)

% = 0.45224742004106549850654336483224793417

? SumEulerrat(1 / (x^2 + 1))

% = 0.38905955531696837171041438969249635814

? /* SumEulerrat(1 / (x^(3/2) + 1)) would be illegal, so: */

? SumEulerrat(1 / (x^3 + 1), 1/2)

% = 0.71426973554924155313504391134630993732

5.2. Euler Products

A completely similar method can be used to compute Euler products numeri-
cally. The result is as follows:

Corollary 5.2.1. Let B(x) = 1 +
∑

m≥1 b(m)xm be a power series with con-

stant term 1 with nonzero radius of convergence r, and assume that B(z) 6= 0 for

|z| < r. Let v be the valuation of B(x)−1 and set P (B; s) =
∏

p∈P B(1/ps). Write

log(B(x)) =
∑

m≥1 a(m)xm and define as above

c(n) =
∑

d|n

µ(d)

d
a
(n

d

)

.

Then for all N ≥ 1 and ℜ(s) > max(− log(r)/ log(N), 1/v) we have

P (B; s) =
∏

p∈P, p≤N

B(1/ps)
∏

n≥1

ζ>N (ns)c(n) .

In particular, if b(1) = 0 and r > 1/2 we have

P (B; 1) =
∏

p∈P, p≤N

B(1/p)
∏

n≥2

ζ>N (n)c(n) .

Furthermore c(n) satisfies the recurrence

c(n) = b(n)− 1

n

∑

1≤k≤n−1

kc(k)
∑

1≤q≤n/k

b(n− qk) .

Proof. The first part of the corollary is immediate by applying the preceding
corollary to A(x) = log(B(x)), which has the same valuation as B(x) − 1, and at
least radius of convergence r since we assume that B has no zeros in |z| < r. For
the recurrence, we note that

B′(x) =
∑

m≥1

mb(m)xm−1 = B(x) log(B(x))′ = B(x)
∑

m≥1

ma(m)xm−1 ,

so the recurrence follows by identification of coefficients and the formula
∑

d|n dc(d) =

na(n) which defines c(n). �

Remarks 5.2.2. (1) The coefficients c(n) are the unique exponents such
that we have the formal expansion

1 +
∑

m≥1

b(n)xm =
∏

n≥1

(1− xn)−c(n) .

(2) It is usually preferable to use the formula giving c(n) in terms of a(n).
However it may happen that a(n) is not easy to compute directly, and one
can then use the recurrence for c(n).

A possible program is as follows, essentially copied from the preceding one:

ProdEulerrat.gp

1 /* Compute prod_{p >= a, prime} F(p^s), F rational function. */

2 ProdEulerrat(F, s = 1, a = 2) =

3 { my(B = getlocalbitprec(), vx = variable(F));

4 my(FIm1, rs, N, r, lim, vF, sal, S);

5

6 FIm1 = subst(F, vx, 1/vx) - 1;

7 vF = valuation(FIm1, vx);

8 rs = real(s);

9 if (rs <= 1/vF, error("real(s) <= 1/v"));

10 localbitprec(32);

11 r = 1 / max(polrootsbound(numerator(F)),

12 polrootsbound(denominator(F)));

13 N = ceil(max(30, a+1));

14 if (rs <= -log(r)/log(N), error("real(s) too small"));

15 lim = ceil(B * log(2) / log(N^rs * r)) + 1;

16 localbitprec(B);

17 sal = log(1 + bitprecision(1., B + 32)*FIm1 + O(vx^(lim+1)));

18 bernvec(floor((lim * s + 1) / 2)); /* cache the B_{2n} */

19 S = sum(n = vF, lim,

20 my(m = Sdmob(sal,n)); if (m, m * LogzetaN(n*s, N)));

21 S = exp(S);

22 forprime (p = a, N, S *= subst(F, vx, p^s));

23 return (bitprecision(S, B));

24 }

Examples (again essentially instantaneous):

? ProdEulerrat(1 + 1 / x^3) - zeta(3) / zeta(6)

% = 0.E-38

? ProdEulerrat(1 - 1 / (x - 1)^2, 1, 3) /* Twin prime constant */

% = 0.66016181584686957392781211001455577843

Note that both of these programs are already implemented in Pari/GP under
the respective names sumeulerrat and prodeulerrat.

5.3. Variants Involving log(p) or log(log(p))

A large number of similar sums or products over primes can be computed in
an analogous manner. A simple example is

∑

p∈P log(p)/ps for ℜ(s), for which we

can modify Proposition 5.1.1 essentially by replacing log(ζ(s)) = −∑p∈P log(1 −
p−s) by its derivative ζ′(s)/ζ(s) = −

∑

p∈P log(p)p−s/(1 − p−s). In this way we
immediately compute for instance that

∑

p∈P

log(p)

p2
= 0.4930911093687644621978262 · · ·

We can also compute limits such as

lim
s→1+

(

∑

p∈P

1

ps
− log(ζ(s))

)

= −0.31571845205389007685 · · · ,

simply by suppressing the term m = 1 in the formula expressing S(s) in terms of
log(ζ(ms)), and

lim
x→∞

(

∑

p∈P
p≤x

1

p
− log(log(x))

)

= 0.261497212847642783755 · · · ,

using the easily proved formula

lim
x→∞

(

∑

p∈P
p≤x

1

p
− log(log(x))

)

= γ + lim
s→1+

(

∑

p∈P

1

ps
− log(ζ(s))

)

,

where as usual γ is Euler’s constant.

A little more challenging is the computation of sums over primes p when log(p)
is in the denominator of the summand, typically sums like T (s) =

∑

p∈P 1/(ps log(p))

for ℜ(s) ≥ 1. When log(p) is on the numerator, we have to use the derivative of
log(ζ(s)). Analogously, here we have to use the integral of log(ζ(s)), in other words

∫ ∞

s

log(ζ(t)) dt .

Even though there are faster methods for computing this, for simplicity we use
the doubly-exponential method, and to compute say, the 30 values of this integral
for integer s ∈ [1, 30] to 38 decimals requires only 0.81 seconds. The following
simple-minded program computes T (s), using the formula given in the proposition
below:

SumEulerlog.gp

1 /* Compute sum_{p prime} 1/(p^s log(p)). */

2 SumEulerlog(s, N = max(2, 30/abs(s))) =

3 { my(B = getlocalbitprec(), S = 0, LN, T, lim);

4

5 localbitprec(32); LN = log(N);

6 lim = ceil(B*log(2)/LN);

7 localbitprec(B + 32);

8 forprime(p = 2, N, S += 1 / (p^s * log(p)));

9 LN = bitprecision(LN, B + 32);

10 T = intnuminit(0,[oo, 1]);

11 forsquarefree(K = 1, lim,

12 my([k] = K, m = moebius(K) / k, a = 1 / (k * LN));

13 /* Tk = intnuminit(0, [oo, 1/a]) */

14 my(Tk = vector(#T, i, if (i == 1, T[1], T[i] * a)));

15 S += m * intnum(t = 0, oo, LogzetaN(k * (t + s) , N), Tk));

16 return (S);

17 }

In a few seconds we obtain for instance
∑

p∈P

1

p log(p)
= 1.636616323351260868569658 · · ·

∑

p∈P

1

p2 log(p)
= 0.507782187859199318774375 · · ·

Note that the series
∑

1/(p log(p)) barely converges (and as usual is over the irreg-
ular set of primes), so it is remarkable that there is no problem in computing its
sum.

We leave as an exercise for the reader the task of writing a program to compute
∑

p∈P 1/(pa logb(p)) with a > 1 or a = 1 and b ≥ 1, arbitrary a, and arbitrary
integer b.

An even harder problem is when log(log(p)) (or worse) occurs. This can also be
dealt with, but is more technical. The idea is as follows: recall that we have a linear
formula expressing S(s) =

∑

p∈P 1/ps in terms of f(s) = log(ζ(s)). By derivation,

it follows that we also have a formula for S(k)(s) = (−1)k
∑

p∈P (log(p))
k/ps in

terms of f (k)(s). For instance this is how we computed
∑

p∈P log(p)/p2. We now
would like to differentiate with respect to k, since the derivative with respect to k
of log(p)k is log(log(p)) log(p)k, which is exactly what is needed after setting k = 0.
A priori this does not seem realistic, but in fact is quite easy:

Proposition 5.3.1. (1) For ℜ(s) > 0 and ℜ(x) ≥ 1 we have

∑

p∈P

1

px log(p)s
=

1

Γ(s)

∑

k≥1

µ(k)

k

∫ ∞

0

ts−1 log(ζ(k(t + x))) dt .

(2) For ℜ(s) > −1 and ℜ(x) ≥ 1 we have

∑

p∈P

1

px log(p)s
= − 1

Γ(s+ 1)

∑

k≥1

µ(k)

∫ ∞

0

ts
ζ′

ζ
(k(t+ x)) dt .

Proof. (Sketch.) For (1) it is easy to check the absolute convergence of the
series and of the integral. We then proceeds as above for the evaluation of S(s): we
express log(ζ) as a sum over primes thanks to the Euler product and use Möbius
inversion. For (2), change s into s + 1 and take the derivative with respect to x.
The details of these proofs are left to the reader. �

Using this proposition, one can then compute the Taylor series expansion of
the left hand side around s = 0 for instance, and obtain hundreds of decimals of
expressions such as

∑

p∈P

log(log(p))

p log(p)

and

lim
x→∞

(

∑

p∈P
p≤x

log(log(p))n

p
− log(log(x))n+1

n+ 1

)

.

All these computations are left as exercises for the reader. For instance, we can
write the following program, which is a simple modification of the previous one:

SumEulerloglog.gp

1 /* Compute sum_{p prime} log(log(p))/(p^s log(p)). */

2 SumEulerloglog(s, N = max(2, 30/abs(s))) =

3 { my(B = getlocalbitprec(), S = 0, LN, T, lim, E);

4

5 localbitprec(32); LN = log(N);

6 lim = ceil(B*log(2)/LN);

7 localbitprec(B + 32);

8 forprime(p = 2, N, S += log(log(p)) / (p^s * log(p)));

9 LN = bitprecision(LN, B + 32);

10 T = intnuminit(0,[oo, 1]);

11 E = Euler();

12 forsquarefree(K = 1, lim,

13 my([k] = K, m = moebius(K) / k, a = 1 / (k * LN));

14 /* Tk = intnuminit(0, [oo, 1/a]) */

15 my(Tk = vector(#T, i, if (i == 1, T[1], T[i] * a)));

16 S -= m * intnum(t = 0, oo, (log(t) + E)

17 * LogzetaN(k * (t+s) , N), Tk));

18 return (S);

19 }

In a few seconds we compute that

∑

p∈P

log(log(p))

p log(p)
= 0.6410802156599846604833518891513999518913451 · · ·

Once again we leave as an exercise for the reader the task of writing a program
to compute

∑

p∈P logk(log(p))/(pa logb(p)) with a > 1 or a = 1 nd b ≥ 1, arbitrary
a, arbitrary integer b, and arbitrary nonnegative integer k.

5.4. Variants Involving Quadratic Characters

We now would like to compute prime sums and Euler products which, in addi-
tion to involving regular functions over primes, also involve a quadratic character
(

D
n

)

where D is a not necessarily fundamental discriminant.
A natural idea is to introduce, in addition to the Riemann ζ function, the

Dirichlet L-functions L(χ, s) for Dirichlet characters χ, in particular for χ(n) =
(

D
n

)

.
We can easily generalize the results of the previous sections to this case. If for

any character χ we set

S(χ, s) =
∑

p

χ(p)

ps

then, exactly as before and with evident notation Möbius inversion gives

S(χ, s) =
∑

p∈P, p≤N

χ(p)

ps
+
∑

m≥1

µ(m)

m
log(L>N(χm,ms)) .

We will briefly study below the computation of L>N (χm,ms).
Let us specialize to the case where χ is a quadratic character

(

D
n

)

for some (not

necessarily fundamental) discriminant D. Then χk = χ if k is odd and χk = χ0,

the trivial character modulo D (χ0(n) = 1 if (n,D) = 1 and χ0(n) = 0 otherwise)
if k ≥ 2 is even. Thus,

S(χ, s) =
∑

χ(p)=1

1

ps
−

∑

χ(p)=−1

1

ps
,

hence
∑

χ(p)=1

1

ps
=

1

2
(S(χ0, s) + S(χ, s)) and

∑

χ(p)=−1

1

ps
=

1

2
(S(χ0, s)− S(χ, s)) ,

where

S(χ0, s) = S(s)−
∑

p|D

1

ps

with the notation of the previous section.
For j = −1, 0, 1 corresponding to the three possible values of χ, let Aj(x) =

∑

m≥1 aj(m)xm be power series satisfying the same assumptions as in Corollary
5.1.2. We set

S(A; s) =
∑

j∈{−1,0,1}

∑

p∈P, χ(p)=j

Aj(1/p
s) .

A generalization of Corollary 5.1.2, which is an immediate consequence of the above
formulas is as follows:

Proposition 5.4.1. Define

cj(n) =
∑

d|n, 2∤d

µ(d)

d
aj

(n

d

)

.

Then for all N ≥ 1 and ℜ(s) > max(− log(r)/ log(N), 1/v) we have

S(A; s) =
∑

j∈{−1,0,1}

∑

p∈P, p≤N χ(p)=j

Aj(1/p
s)

+
1

2

∑

n≥1

(c1(n)− c−1(n)) log(L>N(χ, ns))

+
1

2

∑

n≥1

(c1(n) + c−1(n)− c1(n/2)) log(ζp∤D, >N (ns))

+
∑

p|D, p>N

A0(1/p
s) ,

with the usual convention that c1(x) = 0 if x /∈ Z and where we set

ζp∤D, >N (s) = L>N(χ0, s) = ζ>N (s)
∏

p|D, p>N

(

1− 1

ps

)

.

It is clear that such formulas can be generalized to Dirichlet characters χ of
higher order r, and the result involves in a simple manner the functions L(χb, N)
for 0 ≤ b < r. The quadratic case is especially simple since it involves L(χ,N) and
L(χ0, N) = L(χ0, N) which is essentially the Riemann ζ function.

We must also explain how to compute the quantities L>N(χ, ns), or equiv-
alently L(χ, ns), at least when χ =

(

D
.

)

. When |D| is very small (for instance
D = −3 or D = −4), we can use the variant of the sumalt algorithm explained to
us by B. Allombert and described in Section 4.5.3.

When ℜ(ns) is large and we do not need too much accuracy, we can also
directly use the definition (either as a sum or as an Euler product). Otherwise,
we can trivially reduce to the case of a fundamental discriminant D, and we can
use one of several methods for computing L(χ, s) such as the one described in
Chapter 9, which in GP is simply lfun(D,n*s) (note that the lfun command can
be used with general Dirichlet characters, not only quadratic, and vastly more
general types of L-functions; however, it does require the existence of a functional
equation of standard type, which explains the necessity in our situation of reducing
to fundamental discriminants).

5.5. Variants Involving Congruences

Let a and k be coprime positive integers. Instead of considering Euler sums
or products over all primes, we can consider the same but restricted to primes
congruent to a modulo k. This is easily done as follows. For all a coprime to k and
s with ℜ(s) > 1 we define

Sa(s) =
∑

p∈P, p≡a (mod k)

1

ps
and T (χ, s) =

∑

a mod k

χ(a)Sa(s) .

For any character χ modulo k we have

log(L(χ, s)) =
∑

p∈P

− log

(

1− χ(p)

ps

)

=
∑

m≥1

1

m

∑

p∈P

χm(p)

pms

=
∑

m≥1

1

m

∑

a mod k

χm(a)Sa(ms) =
∑

m≥1

T (χm,ms)

m
.

It follows that for any integer n ≥ 1

log(L(χn, ns))

n
=
∑

m≥1

T (χmn,mns)

mn
,

so by the second Möbius inversion formula we have in particular

T (χ, s) =
∑

a mod k

χ(a)Sa(s) =
∑

m≥1

µ(m)

m
log(L(χm,ms)) .

Now by orthogonality of characters, if we denote by Xk the group of characters of
(Z/kZ)∗ we have for b coprime to k

∑

χ∈Xk

χ(b)T (χ, s) =
∑

a mod k

Sa(s)
∑

χ∈Xk

χ(ab−1) = φ(k)Sb(s) ,

hence

Sa(s) =
1

φ(k)

∑

χ∈Xk

χ(a)
∑

m≥1

µ(m)

m
log(L(χm,ms)) .

As above, it is then immediate to compute any convergent Euler sum of the form
∑

p≡a (mod k) f(p) or any convergent Euler product of the form
∏

p≡a (mod k) f(p).

As mentioned above, to compute L(χn, ns) (hence L>N (χn, ns)) the simplest
is to use the preprogrammed Pari/GP lfun program.

As an application, if we set Pa =
∏

p≡a (mod 5) 1/(1 − 1/p2), we immediately

compute that

P1 = 1.0109151606010195226049565842895149209845386275817385237 · · ·
P2 = 1.3685720538766490858607638904831099901702078288858952050 · · ·
P3 = 1.1357648786689216268686430094720822895119364130054687441 · · ·
P4 = 1.0049603239222975589937496248102521847955102941880228801 · · · .

Of course P1P2P3P4 = (1 − 1/52)ζ(2) = 4π2/25.
The product P2P3 was computed in [ERS19] using a different and more com-

plicated method which apparently does not allow the computation of P2 and P3

individually, however see [Ram19].

5.6. Hardy–Littlewood Constants: Quadratic Polynomials

Let A(X) ∈ Z[X ] be a polynomial with integer coefficients and assume that
A(X) is irreducible in Q[X ] and has content 1. For any prime p, we let ω(p) =
ω(A, p) be the number of solutions in Fp of A(x) = 0. Then conjecturally, the
number of integers n with 1 ≤ n ≤ N such that |A(n)| is prime should be asymptotic
to

H(A)

deg(A)
· N

log N
, where H(A) =

∏

p

p− ω(p)

p− 1

is the so-called Hardy–Littlewood constant of the polynomial A. It is therefore
interesting to compute this Euler product. In particular, polynomials with a large
H(A) should have asymptotically more prime values.

The case of linear polynomials being trivial (H(aX+b) = a/φ(a) when gcd(a, b) =
1), in the present section we first treat the next simplest case of quadratic polynomi-
als, and consider polynomials of larger degree later. Thus, let A(X) = aX2+bX+c
be an irreducible quadratic polynomial with gcd(a, b, c) = 1, and let D = b2 − 4ac
be its discriminant. It is easy to see that ω(p) is given by the following formulas.

• If p ∤ a, then ω(p) = 1 +
(

D
p

)

.

• If p | a and p ∤ b, then ω(p) = 1.
• If p | a and p | b (hence p ∤ c), then ω(p) = 0.
A little computation left to the reader shows the following:

Proposition 5.6.1. Let A(X) = aX2 + bX + c be an irreducible polynomial of

degree 2 with gcd(a, b, c) = 1. The Hardy–Littlewood constant of A is given by the

formula

H(A) = c2
∏

p>2



1−

(

D
p

)

p− 1





∏

p|a, p∤2b

p− 1

p− 2

∏

p|gcd(a,b), p>2

p

p− 1
,

where c2 = 0 if c is even and a+ b is even, c2 = 1 if a+ b+ c is odd, and c2 = 2 if

c is odd and a+ b is even.

Write D = D0f
2 with D0 a fundamental discriminant. Since

∏

p>2



1−

(

D
p

)

p− 1



 =
∏

p>2



1−

(

D0

p

)

p− 1





∏

p|D, p>2



1−

(

D0

p

)

p− 1





−1

,

we must therefore compute the Euler product

C(D0) =
∏

p>2



1−

(

D0

p

)

p− 1





for a fundamental discriminant D0. With a slight abuse of notation, set C(D0, s) =
∏

p>2(1 −
(

D0

p

)

/(ps − 1)), so that C(D0) = C(D0, 1), and set χ(n) =
(

D0

n

)

. Then

for ℜ(s) > 1

− log(C(D0, s)) = −
∑

p, χ(p)=1

log
ps − 2

ps − 1
−

∑

p, χ(p)=−1

log
ps

ps − 1
,

hence with the notation of Section 5.4, C(D0, s) = e−S(A;s) with A0(1/p
s) = 0,

A1(1/p
s) = − log

(

ps − 2

ps − 1

)

=
∑

m≥1

2m − 1

mpms
,

A−1(1/p
s) = − log

(

ps

ps − 1

)

= −
∑

m≥1

1

mpms
,

so that a1(m) = (2m − 1)/m and a−1(m) = −1/m. Thus,

c1(n) =
1

n

∑

d|n, 2∤d

µ(d)(2n/d − 1)

and

c−1(n) = − 1

n

∑

d|n, 2∤n

µ(d) .

Set a(n) = (c1(n)− c−1(n))/2 and b(n) = (c1(n) + c−1(n)− c1(n/2))/2, so that

S(A; s) =
∑

n≥1

a(n) log(L(χ, ns)) +
∑

n≥1

b(n) log(ζp∤D0
(ns)) .

Then by the above we have

a(n) =
1

2n

∑

d|n, 2∤d

µ(d)2n/d

and one easily checks that b(n) = a(n) − a(n/2) if n > 1, while b(1) = 0 (recall
that we set a(x) = 0 if x is not integral), so we can make s → 1 by limiting the
second sum to n ≥ 2. We can thus compute C(D0), using as above Lp>N(χ, n) and
ζp∤D0, p>N (n). The following program implements this:

HardyLittlewood2.gp

1 /* Auxiliary functions. */

2 ZetaDN(P, s) = zeta(s) * prod(j = 1, #P, 1 - P[j]^(-s));

3

4 LchiN(L, Ebad, s) =

5 { my([P, E] = Ebad);

6 lfun(L, s) * prod(j = 1, #P, subst(E[j], ’x, P[j]^(-s)));

7 }

8

9 LchiNinit(D, P) =

10 { my(Ebad = [], Pbad = []);

11 foreach (P, p,

12 my(s = kronecker(D, p));

13 if (s, Ebad = concat(Ebad, 1 - s*’x);

14 Pbad = concat(Pbad, p)));

15 return ([Pbad, Ebad]);

16 }

17

18 Oddpart(n) = n >> valuation(n,2);

19

20 /* The real work; D is fundamental. */

21 HLW2(D, N) =

22 { my(B = getlocalbitprec(), lim, S1, S2, L, P, v, Ebad);

23

24 localbitprec(32); lim = ceil(B*log(2)/log(N/2));

25 localbitprec(B + lim + exponent(lim));

26 L = lfuninit(D, [1/2, lim, 0]);

27 v = vector(lim);

28 forfactored(X = 1, lim,

29 my([n, fan] = X, S = 0, P = fan[,1]);

30 if (n % 2 == 0, P = P[^1]);

31 X = matconcat([P, vectorv(#P,i,1)]);

32 fordivfactored(X, Y,

33 my([d] = Y); \\ odd squarefree divisor of X

34 S += moebius(Y) << (n/d));

35 v[n] = S / (2*n);

36 );

37 P = setunion(factor(abs(D))[,1]~, primes([2, N]));

38 Ebad = LchiNinit(D, P);

39 S1 = sum(n = 1, lim, v[n] * log(LchiN(L, Ebad, n)));

40 S2 = sum(n = 2, lim, (v[n] - if (n%2 == 0, v[n/2]))

41 * log(ZetaDN(P, n)));

42 return (S1 + S2);

43 }

44

45 /* Compute the Hardy-Littlewood constant of aX^2+bX+c. */

46 HardyLittlewood2(A, N = 50) =

47 { my(D = poldisc(A), S, P);

48

49 if (poldegree(A) != 2, error("polynomial of degree != 2"));

50 my([a, b, c] = Vec(A));

51 if (issquare(D) || gcd([2 * a, a + b, c]) > 1, return (0));

52 N = max(N, 3);

53 /* Take care of the prime p = 2. */

54 S = if ((a + b) % 2, 1., 2.);

55 /* Take care of odd primes dividing a. */

56 P = factor(Oddpart(a))[,1];

57 foreach (P, p,

58 S *= if (b % p, (p - 1) / (p - 2), p / (p - 1))

59 );

60 /* Take care of odd primes dividing the index f. */

61 my([D0, f] = coredisc(D, 1));

62 P = factor(Oddpart(f))[,1];

63 S /= vecprod([1 - kronecker(D0, p) / (p - 1) | p <- P]);

64 /* Take care of the primes p <= N. */

65 S *= prodeuler(p = 3, N, 1 - kronecker(D0, p) / (p - 1));

66 /* Do the real work */

67 return (S * exp(-HLW2(D0, N)));

68 }

For example, using this program we compute that

H(X2 +X + 41) = 6.6395463549428433306471137152997759329371091 . . .

H(X2 +X + 75) = 0.6219533598519743400087125748592568290582438 . . .

H(2X2 − 199) = 7.3291180993696071658232761749275362031861488 . . . .

Thus, we can reasonably expect that the first and third polynomials will produce
many primes (and in some sense are record-breaking), while the second will produce
much fewer primes (but it is easy to find polynomials of the same shape which
produce even less).

Two remarks:

Remarks 5.6.2. (1) The use of the preliminary lfuninit command in
the HLW2 program avoids recomputing some data for each value of n, and
speeds up the program considerably.

(2) Note the use of the forfactored and fordivfactored commands, which
avoid factoring several times the same integers (refer to the Pari/GPman-
ual for the use of these functions).

(3) The bit accuracy B needs to be increased by two increments: first a quan-
tity lim which measures the difference between relative and absolute accu-
racy (GP works with relative, but we need absolute), and second a quantity
exponent(lim) which measures approximately the number of bits of ac-
curacy lost when summing lim terms. In other scripts this additional
quantity was often arbitrarily set to 5, 10, or more often 32, but here
it is essential to optimize since the speed of the lfuninit command is
extremely sensitive to the bit accuracy.

As an amusing exercise, we can search for quadratic polynomials whose Hardy–
Littlewood constant is large, and in view of the above examples, larger than 6, say.
We may of course assume that a > 0, and since changing X into X + 1 changes
b into b + 2a and changing X into −X changes b into −b we may assume that
0 ≤ b ≤ a. This of course only excludes the most basic polynomial transformations.
In addition, since we are looking for large H(A), we may assume that c is odd
(otherwise all even x are excluded), hence that a + b is even (otherwise all odd x

are excluded), and this exactly corresponds to the case c2 = 2 in the formula for
H(A). We can thus write the following:

HardyLittlewoodsearch.gp

1 HardyLittlewoodsearch(lima, limc) =

2 { my(V = List());

3

4 localbitprec(64);

5 if (limc % 2 == 0, limc++);

6 for (a = 1, lima,

7 forstep (b = a%2, a, 2,

8 my(g = gcd(a, b), P = a * ’x^2 + b * ’x);

9 forstep (c = -limc, limc, 2,

10 if (gcd(c, g) > 1, next());

11 my(r = HardyLittlewood2(P + c, 50));

12 if (r > 6.5, listput(V, [[a,b,c], r]); print(V[#V]))

13 )

14 )

15 ); \\ sort with respect to ’r’

16 return (vecsort(Vec(V), 2, 12));

17 }

Note that the flag 12 in the vecsort command means first that we want to
sort by descending instead of ascending order, and second that we want to remove
duplicate entries.

Running this program with lima=12 and limc=1000, we see that in that range
the largest H(A) is indeed obtained for A(X) = 2X2 − 199 (and a few equivalent
polynomials) with H(A) = 7.3291 · · · as given above, followed by A(X) = 6X2 +
6X + 31 (and equivalent) with H(A) = 6.9208 · · · .

On the other hand, by making a less systematic search using a truncated form
of the formula for H(A), it is easy to find quadratic polynomals with H(A) > 8,
for instance H(37X2 +23X − 8863) = 8.097818 · · · . In any case, it is easy to show
that H(A) is unbounded.

5.7. Hardy–Littlewood Constants: General Polynomials

5.7.1. Cubic Polynomials. Let A(X) = aX3 + bX2 + cX + d be a cubic
polynomial. In order for A to represent infinitely many primes it is evidently
necessary that A be irreducible and that gcd(a, b, c, d) = 1. But writing

A(X) = 6a(X(X − 1)(X − 2)/6) + (6a+ 2b)(X(X − 1)/2) + (a+ b+ c)X + d

we see that a stronger condition is gcd(6a, 2b, a + b + c, d) = 1 (similar to the
condition gcd(2a, a+b, c) = 1 used in the previous script). LetD be the discriminant
of the polynomial A. The primes p dividing a or D are finite in number, so for such
primes ω(p) can be computed directly (we will see below the GP commands). Thus,
let p be a prime not dividing a or D, and let K be the cubic number field defined
by the polynomial A. Since p ∤ aD the decomposition of the polynomial A modulo
p reflects the decomposition of the prime p in the extension K/Q, hence ω(p) is
equal to the number of prime ideals of K above p which are of residual degree 1.

In other words, with evident notation, if p splits as 3, 21, or 111 then ω(p) = 0, 1,
or 3 respectively. Thus again with evident notation we have

H(A) =
∏

p|aD

p− ω(p)

p− 1
(P3P111)(1) with

P3(s) =
∏

p∤aD
pZK=p3

ps

ps − 1
and P111(s) =

∏

p∤aD
pZK=p1p

′

1p
′′

1

ps − 3

ps − 1
.

Thus log(P3(s)) =
∑

k≥1 S3(ks)/k and log(P111(s)) = −∑k≥1(3
k − 1)S111(ks)/k

with S3(s) =
∑

p∤aD, pZK=p3
1/ps and similarly for S111(s).

We must now distinguish the Galois type of the number field K. Assume first
that K is a cyclic cubic field, i.e., that D is a square. This case is essentially
identical with the quadratic case, with 2 replaced by 3 (this is true more generally
for Cℓ-fields with ℓ prime). More precisely, we have

S3(s) =
1

3

∑

3∤n

µ(n)

n
log((ζ3/ζK)p∤aD(ns))

S111(s) =
∑

n≥1

µ(n)

n
log(ζp∤aD(ns))− S3(s) .

Thus, if we set

c1(n) =
1

n

∑

d|n

µ(d)(3n/d − 1) and c2(n) =
1

3n

∑

d|n, 3∤d

µ(d)3n/d

we have

log((P3P111)(s)) = −
∑

n≥1

(c1(n) log(ζp∤aD(ns)) + c2(n) log((ζ
3/ζK)p∤aD(ns)))

=
∑

n≥1

a(n) log((ζK/ζ)p∤aD(ns)) +
∑

n≥1

c(n) log(ζp∤aD(ns))

with a(n) = −c2(n) and c(n) = 2c2(n) − c1(n). Since c1(1) = 2 and c2(1) = 1, we
have c(1) = 0 (otherwise our Euler product would not converge), and for n ≥ 2
we have c1(n) = (1/n)

∑

d|n µ(d)3
n/d. An easy computation shows that c1(n) =

3c2(n)−c2(n/3), so that c(n) = −c2(n)+c2(n/3) = a(n)−a(n/3), exactly analogous
to the quadratic case.

Assume now that K is a noncyclic cubic field, and let k its quadratic resolvent
field, in other words k = Q(

√
D). We first note that it follows from the quadratic

case that (with evident notation)

S21(s) =
1

2

∑

2∤n

µ(n)

n
log((ζ2/ζk)p∤aD(ns))

(although a priori we do not need S21, we will need it for S111). A completely
similar computation gives

S3(s) =
1

3

∑

3∤n

µ(n)

n
log((ζζk/ζK)p∤aD(ns)) ,

hence

S111(s) =
∑

n≥1

µ(n)

n
log(ζp∤aD(ns))− S21(s)− S3(s) .

Thus, if in addition to c1 and c2 defined above we set

c3(n) =
1

2n

∑

d|n, 2∤d

µ(d)(3n/d − 1) ,

we have

log((P3P111)(s)) = −
∑

n≥1

(c1(n) log(ζp∤aD(ns))

+ c2(n) log((ζζk/ζK)p∤aD(ns)) + c3(n) log((ζ
2/ζk)p∤aD(ns)))

=
∑

n≥1

a(n) log((ζK/ζ)p∤aD(ns))+

∑

n≥1

b(n) log((ζk/ζ)p∤aD(ns)) +
∑

n≥1

c(n) log(ζp∤aD(ns)) ,

with a(n) = −c2(n), b(n) = c2(n)−c3(n), and c(n) = −c1(n)+c2(n)+c3(n), and As
usual c1(1) = 2, c2(1) = 1, and c3(1) = 1, so c(1) = 0 as it should for convergence.
As above, since c1(n) = 3c2(n)−c2(n/3) we check that c(n) = a(n)−b(n)−a(n/3).

This shows that the formulas in the cyclic and noncyclic case can be unified by
setting b(n) = 0 (and ζk/ζ = 1) in the cyclic case.

This leads to the following program:

HardyLittlewood3.gp

1 /* In the cyclic cubic case, we need z^3/z_K=z^2/(z_K/z) */

2 /* In the noncyclic cubic case, we need z^2/z_k=z/L_D

3 and zz_k/z_K=z_k/(z_K/z)=zL_D/(z_K/z). */

4

5 /* Auxiliary functions. */

6 /* ZetaDN and LchiN are defined in Hardylittlewood2.gp. */

7

8 LchiKNinit(nf, P) =

9 {

10 my(Ebad = [], Pbad = [], D = nf.disc);

11 foreach (P, p,

12 my(E, v = idealprimedec(nf, p));

13 E = if (#v == 1, if (D % p == 0, next);

14 1 + ’x + ’x^2,

15 #v == 2, if (D % p == 0, 1 - ’x, 1 - ’x^2),

16 #v == 3, (1 - ’x)^2);

17 Ebad = concat(Ebad, E);

18 Pbad = concat(Pbad, p);

19 );

20 return ([Pbad, Ebad]);

21 }

22

23 HLW3(A, N, P, D) =

24 { my(flnoncyc = (D != 1), nf, lim);

25 my(LK, va, vb, vc, S1, S2, S3);

26 my(B = getlocalbitprec(), EKbad);

27

28 localbitprec(32);

29 lim = ceil(B*log(2)/log(N/3)); nf = nfinit(A);

30 localbitprec(B + ceil(1.585*lim) + exponent(lim));

31 LK = lfundiv(lfuncreate(A), lfuncreate(1));

32 LK = lfuninit(LK, [1/2, lim, 0]);

33 va = vector(lim); vb = vector(lim); vc = vector(lim);

34 forfactored(X = 1, lim,

35 my([n] = X);

36 S2 = S3 = 0;

37 fordivfactored(X, Y,

38 my([d] = Y, n3 = 3^(n/d), mob = moebius(Y));

39 if (mob,

40 if (d % 2 && flnoncyc, S2 += mob * (n3 - 1));

41 if (d % 3, S3 += mob * n3)

42 )

43 );

44 va[n] = -S3 / (3*n);

45 if (flnoncyc, vb[n] = -(va[n] + S2 / (2*n)));

46 vc[n] = va[n] - vb[n] - if (n%3 == 0, va[n/3], 0);

47 );

48 EKbad = LchiKNinit(nf, P);

49 S1 = sum(n = 1, lim, va[n] * log(LchiN(LK, EKbad, n)));

50 S2 = 0;

51 if (flnoncyc,

52 my(LD, Echibad = LchiNinit(D, P));

53 LD = lfuninit(D, [1/2, lim, 0]);

54 S2 = sum(n = 1, lim, vb[n] * log(LchiN(LD, Echibad, n)))

55 );

56 S3 = sum(n = 2, lim, vc[n] * log(ZetaDN(P, n)));

57 return (S1 + S2 + S3);

58 }

59

60 /* Compute the Hardy-Littlewood constant of aX^3+bX^2+cX+d. */

61 HardyLittlewood3(A, N = 50) =

62 { my(DA = poldisc(A), S = 1., Da6, P, v = variable(A));

63

64 if (poldegree(A) != 3, error("polynomial of degree != 3"));

65 my([a, b, c, d] = Vec(A));

66 if (!polisirreducible(A) ||

67 gcd([6 * a, 2 * b, a + b + c, d]) > 1, return (0));

68 N = max(N, 5);

69 /* Take care of bad primes and p <= N. */

70 Da6 = abs(6 * a * DA);

71 P = setunion(factor(Da6)[,1]~, primes([5, N]));

72 foreach (P, p, S *= (p - #polrootsmod(A, p)) / (p - 1));

73 /* Do the real work. */

74 if (a != 1, A = a^2 * subst(A, v, v / a));

75 return (S * exp(HLW3(A, N, P, coredisc(DA))));

76 }

77

78 HardyLittlewood(A, N = 50) =

79 { my(d = poldegree(A));

80

81 if (d <= 0, return (0),

82 d == 1,

83 my(a = abs(polcoeff(A, 1))); return (a / eulerphi(a)),

84 d == 2, return (HardyLittlewood2(A, N)),

85 d == 3, return (HardyLittlewood3(A, N)),

86 error("HardyLittlewood not implemented for d >= 4")

87 );

88 }

This program illustrates a number of important GP commands:

(1) As in the quadratic case, we need the lfun command, but now not only for
ζk/ζ = L(χD) (obtained in the function LchiN), but also for the nontrivial
ζK/ζ obtained as follows: lfuncreate(A) creates ζK , lfuncreate(1)

creates ζ in the necessary format for lfun functions, and finally lfundiv

performs the division ζK/ζ.
(2) Note the use of #polrootsmod(A, p) which computes ω(p) directly, but

is used only for a small number of primes (those dividing 6aD and those
less than or equal to N).

(3) Note the use of #idealprimedec(nf, p) which allows the computation
of the Euler factor of ζK/ζ at these same primes, using the function
LchiKNEulerbad to determine this factor for the 5 possible types of prime
decomposition in K. In fact, by a basic result of algebraic number theory,
for the primes p not dividing Da6, we could use #idealprimedec(nf, p)

instead of #polrootsmod(A, p) (of course, idealprimedec is necessary
for the Euler factors).

5.7.2. Cℓ Polynomials. What we did in the cyclic cubic case (not in the S3

case) can immediately be generalized to the cyclic Cℓ case for ℓ prime, generalizing
both the quadratic and the cyclic cubic case.

Let A(X) =
∑

0≤n≤ℓ anX
n be an irreducible polynomial of degree exactly equal

to ℓ, and assume that the number field K defined by A is abelian with Galois group
Cℓ. We can immediately copy what we did above. First, setting D(A) = disc(A)
and E(A) = ℓ!aℓD(A), with evident notation we have

H(A) =
∏

p|E(A)

p− ω(p)

p− 1
(PℓP1ℓ)(1) with

Pℓ(s) =
∏

p∤E(A)
pZK=pℓ

ps

ps − 1
and P1ℓ(s) =

∏

p∤E(A)
pZK=p1p

′

1···

ps − ℓ

ps − 1
,

and performing exactly the same computation we find that if we set

a(n) = − 1

ℓn

∑

d|n, ℓ∤d

µ(d)ℓn/d

and c(n) = a(n)− a(n/ℓ), we have

log((PℓP1ℓ)(s)) =
∑

n≥1

a(n) log((ζK/ζ)p∤E(A)(ns)) +
∑

n≥2

c(n) log(ζp∤E(A))(ns) .

We leave to the reader the immediate task of writing the corresponding GP script
which will be essentially identical to the cyclic cubic one.

5.7.3. General Polynomials. The case of general polynomials of degree d ≥
4 is more complicated, and it is not completely clear how to treat them, at least
in a straightforward manner. As in the quadratic and cubic case, we can evidently
reduce to the computation of what one can call the Hardy–Littlewood constants of
number fields (as opposed to polynomials). When the number field is abelian over
Q, there is no problem since we have d Artin L-functions which are simply Hecke
L-series, and the splitting of the primes is completely understood in terms of these
series (we have mentioned the case of Cℓ-number fields above, when ℓ is prime).

If it is not abelian, or even Galois, it is not clear to the authors how to proceed
in general, although it may not be too difficult. What we explained above for S3-
cubic fields can also easily be done for A4-quartic fields, but it is not completely
clear how to do it for D4- or S4-quartic fields. We leave to the reader to explore
this avenue of computation.


